

(51) 4042-2828 (FIXO). (51) 99602-6897 (VIVO e WhatsApp)

APRESENTAÇÃO

Você adquiriu um produto fabricado no Brasil. Muito obrigado incentivar e fortalecer o desenvolvimento econômico e tecnológico nacional!


A placa **Controladora de Relés USB** tem como função básica ligar/desligar os relés a partir de comandos recebidos do computador. Além disso, a placa monitora 4 entradas digitais e informa ao computador o status lógico de cada uma. Neste manual é apresentado o protocolo de comunicação da CONTROLADORA e o modo da conexão elétrica dos relés para que você possa desenvolver o seu software.

Cada relé é controlado por um dado específico vindo do computador, podendo este dado ligar ou desligar o respectivo relé; para ler as entradas digitais da placa o computador deve fazer a solicitação de leitura e em seguida aguardar o dado que será recebido, que é o status da entrada solicitada para leitura. Os sensores que podem ser conectados à CONTROLADORA são: botoeira de contato simples, chave fim de curso, sensores de contato seco em geral, sensores com sinais de 3V a 24V.

ESPECIFICAÇÕES TÉCNICAS

- Conectividade: USB
- Velocidade de comunicação (baudrate): 300 bp/s
- Tipo de Entrada de pulso: Botoeira; Sensor NPN ou PNP; Contato Seco; Pulso de tensão 3V a 24V
- Tipo de Saída / Acionamento: RELÉ com contatos COMUM/NA/NF
Corrente máxima: 7A, Tensão máxima: 220V
Vida útil de cada relé: 300 mil a 3 milhões de comutações
- Alimentação: USB
(A alimentação e os dados são fornecidos através da porta USB)
- Dimensões: Altura: 10cm; Largura: 13cm; Profundidade: 2,5cm

CONECTIVIDADE



INSTALAÇÃO DO DRIVER DA PLACA

Para que a placa seja reconhecida no sistema operacional Windows (todas as versões) é necessário **primeiramente**:


- baixar o driver em www.digitaq.com.br e na guia PRODUTOS localize o seu produto e clique no botão “Driver do Adaptador USB/Serial”.

- Será baixado em seu computador o arquivo compactado em WinRar: **Todos_Drivers_Adaptador_USB_Serial.rar**
- Dê um duplo clique esquerdo no arquivo baixado e instale todos os drivers contidos no pacote de instalação. Pode ser necessário reiniciar o sistema.
- Conecte o cabo USB na CONTROLADORA e também na entrada USB do seu computador ou notebook.

Abra o Gerenciador de dispositivos de seu computador para conferir se o driver foi instalado corretamente (é necessário que a CONTROLADORA esteja conectada ao computador).

Localize a guia “Portas (COM e LPT) verifique se consta o dispositivo **Prolific-USB-to-Serial**

Isto significa que o driver está instalado corretamente e a placa foi detectada pelo Windows.



Na imagem de exemplo acima o Windows especificou que a CONTROLADORA está conectada na porta USB “COM11”, portanto sua aplicação deverá acessar essa porta. Se você conectar a CONTROLADORA em outra porta USB ou em outro computador o Windows poderá determinar uma porta diferente (ex.: COM7). Por isso, há um recurso para informar à sua aplicação em qual porta USB a CONTROLADORA está conectada.

PROTOCOLO DE ACIONAMENTO DOS RELÉS

Nesta secção são apresentados os comandos que podem ser enviados à CONTROLADORA para controlar os relés (o Driver deve ser instalado antes).

- Ajuste a velocidade da comunicação (Baudrate) da sua aplicação em 300 bps (essa velocidade baixa permite o uso de cabos USB longos);
- Selecione a porta COMx detectada pelo seu sistema operacional (Ex.: COM11);

A seguir temos duas tabelas de dados:

A Tabela 1 contendo os caracteres que devem ser enviados para controlar cada relé da CONTROLADORA; Já a Tabela 2 somente tem efeito se você também dispõe da placa de expansão para controlar 4 relés adicionais.


Tabela 1: Controladora Principal	
Comando enviado	Ação na Controladora
1	LIGA RELÉ 1
2	DESLIGA RELÉ 1
3	LIGA RELÉ 2
4	DESLIGA RELÉ 2
5	LIGA RELÉ 3
6	DESLIGA RELÉ 3
7	LIGA RELÉ 4
8	DESLIGA RELÉ 4

Tabela 2: Placa de Expansão	
Comando enviado	Ação na placa de expansão
A	LIGA RELÉ 1
B	DESLIGA RELÉ 1
C	LIGA RELÉ 2
D	DESLIGA RELÉ 2
F	LIGA RELÉ 3
G	DESLIGA RELÉ 3
H	LIGA RELÉ 4
I	DESLIGA RELÉ 4

CONTROLANDO O [RELÉ 1]

Observe que na tabela 1, o **RELE 1** é controlado quando a CONTROLADORA recebe os caracteres “1” e “2”. Se a sua aplicação enviar o caractere “1” via comunicação serial então o **RELÉ 1** irá ligar. Se a aplicação enviar o caractere “2” o **RELÉ 1** irá desligar.

No Visual Basic 6 crie um projeto Standard EXE. Em seguida Clique no menu PROJECT>COMPONENTS e assinale a caixa **Microsoft Comm Control 6.0** e clique no botão OK



Adicione ao formulário:

- 1 objeto MSComm1 (ícone telefone);
- 2 Command Button (Command1 e Command 2);
- 2 Label (Label1 e Label2)
- 1 Timer (Timer1)

Em seguida dê um duplo clique esquerdo no formulário para visualizar a tela de código e no Form_Load() digite:

```
Dim LerEntrada As Integer ' variável global
Private Sub Form_Load()
    'Verifica se o programa e a placa já estão conectados:
    If MSComm1.PortOpen = True Then ' se estiver conectado então
        MSComm1.PortOpen = False ' desconecta o software e a placa
    End If

    'Configura a Porta Serial do computador/notebook

    MSComm1.Settings = "300,n,8,1"
    MSComm1.DTREnable = True
    MSComm1.EOFEnable = False
    MSComm1.Handshaking = comNone
    MSComm1.InBufferSize = 1
    MSComm1.InputLen = 1
    MSComm1.InputMode = comInputModeBinary
    MSComm1.NullDiscard = True
    MSComm1.OutBufferSize = 512
    MSComm1.RThreshold = 1 'Caracteres esperados para gerar a chamada _OnComm
    MSComm1.RTSEnable = True
    Timer1.Interval = 300

    'Tenta abrir desde a COM2 , se houver erro (porta não existir)
    'prossigue tentando abrir a proxima porta até COM30 (2 a 30)
    On Error Resume Next
    For COMNumber = 2 To 30 'tenta abrir desde COM2 até COM30
        MSComm1.CommPort = COMNumber 'define porta COM conforme variavel 'COMNumber'
        MSComm1.PortOpen = True 'tenta abrir a porta COMx
        If MSComm1.PortOpen = True Then
            Label1.Caption = "Conectado em COM" & COMNumber 'Escreve Conectado em COMx
            Exit Sub 'Encerra busca da COMx
        End If
    Next COMNumber
End Sub

Private Sub Command1_Click()
    MSComm1.Output = "1" 'Envia o caractere 1 que liga o relé 1
End Sub

Private Sub Command2_Click()
    MSComm1.Output = "2" 'Envia o caractere 2 que desliga o relé 1
End Sub

Static Sub MSComm1_OnComm()
    Select Case MSComm1.CommEvent
        Case comEvReceive 'Se a controladora enviar um dado ao PC
            Buffer = MSComm1.Input
            If LerEntrada = 0 Then Label2.Caption = StrConv(Buffer, vbUnicode)
        End Select
    End Sub

Private Sub Timer1_Timer()
    LerEntrada = 0
    MSComm1.Output = "E0" 'Solicita status da entrada E0
End Sub
```